

# *Impact of a Planar Kinematic Chain with Granular Matter*

**Seunghun Lee**  
**Dan B. Marghitu**

The theoretical model of a kinematic chain impacting granular matter is studied. The force of the granular medium acting on the chain is a linear superposition of a static (depth-dependent) resistance force and a dynamic (velocity-dependent) frictional force. This resistance force is opposed to the direction of the velocity of the immersed chain. We present two methods (one using `EventLocator` and the other using `FixedStep`) for the problem. As examples, a single and a double pendulum are simulated using different initial impact velocity conditions. We analyze how rapidly the kinematic chain impacting the granular medium slows upon collision. For the analyzed cases the kinematic chain under high impact force (higher initial velocity) comes to rest faster in the granular matter than the same body under low impact force (lower initial velocity).

## ■ Introduction

The physical behavior of granular matter has its own specific properties that are unlike solids, liquids, or gases. Impact with a granular medium is very interesting because granular materials have characteristics similar to a solid, yet flow like a fluid. The study of impact into granular matter is, surprisingly, in its early stages.

The focus of recent work has been to develop a force law for granular impacts and to find a mathematical formula to measure the impact force of objects dropped into granular matter [1]. Tsimring and Volfson [2] studied the penetration of large projectiles into dry granular media. For the resistance-force model, they proposed a drag force depending on the velocity and a friction force depending on the depth. Depth-dependent friction-force models have been developed for horizontal motion in [3, 4, 5, 6] and for vertical motion in [7, 8]. Katsuragi and Durian [1] sparked new interest in the field of impact with granular matter. They applied the resistance-force model proposed in [2] and verified the motion of a rigid

sphere with a line-scan digital CCD camera. They analyzed an interesting phenomenon: how rapidly a sphere impacting a granular medium slows upon collision. Analysis shows that the greater the speed at which the spheres impact the medium, the sooner they will come to rest for the vertical impact.

In our study we focus on modeling and simulating a single and a double pendulum impacting a granular medium using the resistance-force model proposed by Tsimring and Volfson [2] and verified by Katsuragi and Durian [1]. For the single pendulum we apply `NDSolve` with variable step size and the simulation is stopped when the velocity changes sign. This event is captured with the command `EventLocator`. For the double pendulum the equations of motion are solved using `NDSolve` with the `FixedStep` method. In this way we capture the physics of the impact process and we can calculate the final stopping time for the kinematic chain.

## ■ Resistance Forces during Impact

For impact with penetration, the most important interaction between a rigid body and the medium is the resistance force from the moment of impact until the body stops. The resistance forces are composed of a drag force proportional to the square of the velocity and a force related to the plasticity of the medium [9, 10]. For impact of a moving rigid body into a granular medium, recent research [1, 2] shows that the total resistance force acting on the body is the sum of a static resistance force characterized by depth-dependent friction force and a dynamic frictional force characterized by velocity-dependent drag force, that is,  $F_r = F_s(z) + F_d(V)$ , where  $F_s$  is the static resistance force and  $F_d$  is the dynamic frictional force.

The dynamic frictional force  $F_d$ , a velocity-dependent force, is generated by the movement of the rigid body into the granular matter from the moment of impact until rest. Even though the state of the granular matter is not a fluid but a solid, the rigid body moves through the granular matter as if it were partly like a fluid. From this fluid-like behavior, the resistance force is assumed to be like a drag force impeding the motion of the rigid body. Recent results [1, 2] show that  $F_d = -\text{sign}(V) \beta A V^2$ , where  $\beta$  is a drag coefficient determined from experimental results and  $A$  is the reference area. The factor  $-\text{sign}(V)$  is negative because the dynamic frictional force acts in the opposite direction of the velocity  $V$ . The dynamic frictional force can be applied separately, with the horizontal and vertical dynamic frictional forces  $F_{dh}$  and  $F_{dv}$  given by

$$F_{dh} = -\text{sign}(V_h) \beta A_h V_h^2, \quad (1)$$

$$F_{dv} = -\text{sign}(V_v) \beta A_v V_v^2, \quad (2)$$

where  $V_h, V_v$  are the horizontal and vertical velocities of a rigid body and  $A_h, A_v$  are the reference areas to the horizontal and the vertical directions.

The static resistance force is an internal resisting force and appears when an external force is applied. The horizontal static resistance force  $F_{dh}$  is defined as the internal impeding

force acting on the horizontal direction. Recent research results [3, 11] show that it is known to be proportional to the normal component of the pressure acting at the contacting point, which increases linearly with the depth  $z$ , the granular density  $\rho$ , and the gravitational acceleration  $g$ . The horizontal static resistance force of a cylinder including the state of immersion, at any slope, can be generalized as

$$F_{sh} = -\text{sign}(V_h) \eta_h g \rho_g z_T^2 d_c, \quad (3)$$

where  $z_T$  is the depth of the immersed cylinder tip.

The vertical static resistance force is defined as the internal impeding resistance acting on the vertical axis. This force is a nonlinear function of the immersion depth. Hill, Yeung, and Koehler [8] suggest an empirical equation with coefficients calculated from experimental data as

$$F_{sv} = -\text{sign}(V_v) \eta_v (z_T / l)^\lambda g \rho_g U, \quad (4)$$


where  $U$  is the volume of the rigid body and  $l$  is the lateral dimension. The coefficients  $\eta_v$  and  $\lambda$  depend on the shape of the rigid body, the properties of the granular matter and the rigid body, the shape of the medium container, and the direction of motion, such as plunging and withdrawing.

## ■ Application

Equations (1) to (4), representing the resistance forces, contain the sign function of the velocity of the application point of the resistance forces. It is not possible to solve the ordinary differential equations of motion that contain the sign function with the variable step method of `NDSolve`, due to the discontinuity at the origin. We use the `EventLocator` method for `NDSolve` with a variable step to solve the ODE for the single-pendulum impact model. We use the `FixedStep` method to solve the ODE for the double-pendulum impact model. This correctly captures the change in the sign of the velocity.

### □ Single-Pendulum Impact Model

A single-pendulum impact model is presented in Figure 1. The planar motion of the pendulum on impact can be described in a fixed Cartesian coordinate system. The  $x$  axis is horizontal, with the positive sense directed to the right; the  $z$  axis is vertical, with the positive sense directed downward; and the  $y$  axis is perpendicular to the plane of motion. The angle between the  $z$  axis and the pendulum is denoted by  $q$ , the generalized coordinate describing this system.



▲ **Figure 1.** A single-pendulum impact model.

The pendulum has mass  $m$ , length  $L$ , and diameter  $d_c$ . The pendulum impacts a granular medium of density  $\rho$  at an initial impact angle  $q_0$ . Gravitational acceleration is  $g$ . Here are the numerical values for the input data.

$$\begin{aligned} m &= 10; \\ L &= 0.5; \\ d_c &= 0.0254; \\ \rho &= 2.46 \times 10^3; \\ q_0 &= 30 \text{ Pi} / 180; \\ g &= 9.8; \end{aligned}$$

The gravitational force acts at the center of mass C of the pendulum. This is the position vector from the origin O to C.

$$\mathbf{r}_C = \{L / 2 \text{ Sin}[q[t]], 0, L / 2 \text{ Cos}[q[t]]\};$$

The resistance force of the granular media is assumed to act at the point E, the center of the immersed part of the pendulum. The immersed depth is  $z_T$  as shown in Figure 1. The length  $L_E$  is the distance from the origin O to E.

$$\begin{aligned} z_T &= L \text{ Cos}[q[t]] - L \text{ Cos}[q_0]; \\ L_E &= L - 0.5 z_T / \text{Cos}[q[t]]; \end{aligned}$$

These are the position vector and the velocity of the point E.

$$\begin{aligned} \mathbf{r}_E &= \{L_E \text{ Sin}[q[t]], 0, L_E \text{ Cos}[q[t]]\}; \\ \mathbf{V}_E &= \text{Cross}[\mathbf{D}[\{0, q[t], 0\}, t], \mathbf{r}_E]; \end{aligned}$$

This is the gravitational force acting at C.

$$\mathbf{G} = \{0, 0, m g\};$$

We use `NDSolve` to solve the differential equations of motion. The expressions for the resistant forces given by equations (1) to (4) contain the signs of the vertical and horizontal

velocities of the application point E. We want to use a variable step size for the integration for this model, so we cannot use the *Mathematica* function `Sign`. (The function `Sign` can be used with `NDSolve` with the `FixedStep` method; see the double-pendulum example.) Therefore for the case of the single pendulum we introduce the piecewise-constant variables  $\text{sgn}V_h$  and  $\text{sgn}V_v$ , defined in the following lists.

```
data1 = {sgn $V_h$  → -1, sgn $V_v$  → 1,  $\beta$  → 1569.7,  $\eta_h$  → 2.43,
 $\eta_v$  → 10,  $\lambda$  → 1.4};
data2 = {sgn $V_h$  → -1, sgn $V_v$  → -1,  $\beta$  → 1569.7,  $\eta_h$  → 2.43,
 $\eta_v$  → .5,  $\lambda$  → 1.7};
```

The list `data1` is used for plunging ( $V_v > 0$ ) and the list `data2` is used for withdrawing ( $V_v < 0$ ). For the two lists we consider continuous sliding on the horizontal direction. The general coefficients for the resistant force [1, 3, 8] are  $\beta = 1569.7$  and  $\eta_h = 2.43$ . For plunging motion  $\eta_v = 10$ ,  $\lambda = 1.4$  and for withdrawing motion  $\eta_v = 0.5$ ,  $\lambda = 1.7$ . When the pendulum penetrates the granular medium with plunging motion ( $0 < q \leq q_0$ ), the values of  $\text{sgn}V_h$  and  $\text{sgn}V_v$  are  $-1$  and  $1$ . When the pendulum withdraws from the granular medium ( $q < 0$ ), the values of  $\text{sgn}V_h$  and  $\text{sgn}V_v$  are  $-1$  and  $-1$ .

This defines the reference area  $A_h$  and the horizontal dynamic frictional force  $F_{dh}$ .

```
A $_h$  = d $_c$  z $_T$ ;
F $_{dh}$  = -sgn $V_h$   $\beta$  A $_h$  V $_E$  [ [1] ] ^ 2;
```

Here are the reference area  $A_v$  and the vertical dynamic frictional force  $F_{dv}$ .

```
A $_v$  = d $_c$  z $_T$  Abs [ Tan [ q [ t ] ] ];
F $_{dv}$  = -sgn $V_v$   $\beta$  A $_v$  V $_E$  [ [3] ] ^ 2;
```

Here is the horizontal static resistance force.

```
F $_{sh}$  = -sgn $V_h$   $\eta_h$   $\rho$  g d $_c$  z $_T$  ^ 2 ;
```

Here are the immersed volume  $U$  of the pendulum and the vertical dynamic frictional force  $F_{sv}$ .

```
U = z $_T$  / Cos [ q [ t ] ] d $_c$  ^ 2 Pi / 4;
F $_{sv}$  = -sgn $V_v$   $\eta_v$  ( z $_T$  / d $_c$  ) ^  $\lambda$   $\rho$  U;
```

This defines the resistance force.

```
F $_r$  = { F $_{dh}$  + F $_{sh}$ , 0, F $_{dv}$  + F $_{sv}$  };
```

The equation of motion of the pendulum is

$$I_O \ddot{q} = (\mathbf{r}_C \times \mathbf{G} + \mathbf{r}_E \times \mathbf{F}_r) \cdot (0, 1, 0). \quad (5)$$

```

I0 = m / 3 (L ^ 2);
Eq5 = I0 q' '[t] == (Cross[rC, G] + Cross[rE, Fr])[[2]];

```

Here  $I_0$  is the mass moment inertia about the origin O. The ODE, equation (5), is solved in `single`. The input variable `dq` is the initial impact angular velocity of the pendulum. The initial start time `t0`, the initial impact angle `q0`, and the initial impact angular velocity `dq0` are defined as initial values. The interpolating function `sol`, which is the solution of the ODE, is defined as a local symbol. To save the solution data for the angle `q` and for the angular velocity `q'`, we introduce initially the empty vectors `qresults` and `dqresults`. The simulation is performed until the angular velocity `q'` becomes zero (considering a numerical error) using the `While` statement. At the first loop the ODE uses the coefficients list `data1` for the plunging motion. `NDSolve` solves this ODE until the  $z$  directional velocity  $V_{Ez}$  of the point E becomes zero and the event is identified using the `EventLocator` method. This event occurs when `q'` is zero or when  $V_{Ez}$  changes sign at `q = 0`. We do not consider the  $x$  velocity,  $V_{Ex}$ , of the point E because the sign of the velocity  $V_{Ex}$  does not change until the angular velocity `q'` becomes zero. Once `NDSolve` stops solving the ODE, the stopping time is saved as `tend`. The last values of the angle `q` and the angular velocity `q'` of the stopping time, `tend`, are saved as the initial values `q0` and `dq0` for the next loop. The `Table` saves the data of the angle `q` and the angular velocity `q'` using `qresult` and `dqresult`. After saving the data, the new initial start time `t0` of the next loop is defined as the stopping time `tend` and the motion index increases from the plunging motion to the withdrawing motion. After the end of the first loop, the `While` statement checks the last value of the angular velocity and decides whether to do one more loop of calculation or to end the module.

```

single[dq_] := Module[{t0 = 0, q0 = q0, dq0 = dq, sol},
  qresults = {}; dqresults = {}; motion = 1;

  While[Chop[dq0, 10-6] != 0,
    Eq50 = Eq5 /. Which[motion == 1, data1, motion == 2, data2];

    sol = NDSolve[{Eq50, q[t0] == q0, q'[t0] == dq0}, q,
      {t, t0, Infinity}, MaxSteps -> Infinity,
      Method -> {"EventLocator", "Event" -> VE[[3]],
        "EventAction" -> Throw[tend = t; q0 = q[t];
          dq0 = q'[t];, "StopIntegration"],
      "EventLocationMethod" -> "LinearInterpolation",
      "Method" -> "ExplicitEuler"}];

    qresult = Table[{t, q[t] 180 / Pi /. sol[[1]]},
      {t, t0, tend, .0025}];
    dqresult = Table[{t, q'[t] /. sol[[1]]},
      {t, t0, tend, .0025}];

    qresults = Append[qresults, qresult];

```

```

dqresults = Append[dqresults, dqresult];

t0 = tend;
motion++] (* end While *)
] (* end Module *)

```

The simulation of the model is performed for an initial impact angle  $q_0 = 30^\circ$  and for three initial impact angular velocities ( $\dot{q}_0 = -3, -1.5,$  and  $-0.1$  rad/s). The angle  $q$  and the angular velocity  $\dot{q}$  are plotted with the blue line for  $\dot{q}_0 = -3$  rad/s, the red dashed line for  $-1.5$  rad/s, and the black dot-dashed line for  $-0.1$  rad/s.

```

single[-3];
Print["The stopping time for  $\dot{q}_0 = -3$  rad/s is ",
tend, " [s]"];
qplot1 = ListLinePlot[qresults, PlotRange → Full,
PlotStyle → {Directive[Blue]}];
dqplot1 = ListLinePlot[dqresults, PlotRange → Full,
PlotStyle → {Directive[Blue]}];

single[-1.5];
Print["The stopping time for  $\dot{q}_0 = -1.5$  rad/s is ",
tend, " [s]"];
qplot2 = ListLinePlot[qresults, PlotRange → Full,
PlotStyle → {Directive[{Red, Dashed}]}];
dqplot2 = ListLinePlot[dqresults, PlotRange → Full,
PlotStyle → {Directive[{Red, Dashed}]}];

single[-0.1];
Print["The stopping time for  $\dot{q}_0 = -0.1$  rad/s is ",
tend, " [s]"];
qplot3 = ListLinePlot[qresults, PlotRange → Full,
PlotStyle → {Directive[{Black, DotDashed}]}];
dqplot3 = ListLinePlot[dqresults, PlotRange → Full,
PlotStyle → {Directive[{Black, DotDashed}]}];



Show[qplot1, qplot2, qplot3, Frame → True,
FrameLabel → {"time [s]", "q [°]"},
PlotRange → {{0, tend}, {-30, 33}}, GridLines → Automatic]
Show[dqplot1, dqplot2, dqplot3, Frame → True,
FrameLabel → {"time [s]", "q̇ [rad/s]"},
PlotRange → {{0, tend}, {-3.8, .2}}, GridLines → Automatic]

```

The stopping time for  $\dot{q}_0 = -3 \text{ rad/s}$  is 0.430259 [s]

The stopping time for  $\dot{q}_0 = -1.5 \text{ rad/s}$  is 0.474437 [s]


The stopping time for  $\dot{q}_0 = -0.1 \text{ rad/s}$  is 0.553072 [s]



The simulation results of the model for the initial impact angle  $q_0 = 30^\circ$  show that when the initial impact velocity of the pendulum increases, the penetrating angle increases. However, we observed the stopping time, defined as duration from the moment of impact until the angular velocity becomes zero, decreases: for  $\dot{q}_0 = -3 \text{ rad/s}$  the stopping time is 0.430259 s, for  $\dot{q}_0 = -1.5 \text{ rad/s}$  the stopping time is 0.474437 s, and for  $\dot{q}_0 = -0.1 \text{ rad/s}$  the stopping time is 0.553072 s.

## □ Double-Pendulum Impact Model

A double-pendulum impact model is presented in Figure 2. The planar motion of the double-pendulum impact can also be described in a fixed Cartesian coordinate system. The angles between the  $z$  axis and the links are denoted by  $q_1$  and  $q_2$ ; these angles are the generalized coordinates describing this system.



▲ **Figure 2.** A double-pendulum impact model.

The first link of the pendulum has mass  $m_1$  and length  $L_1$ . The second link has mass  $m_2$ , length  $L_2$ , and diameter  $d_c$ . The density of the granular medium is  $\rho$  and the gravitational acceleration is  $g$ . The pendulum impacts a granular medium at the initial impact angles  $q_{10}$  for the first link and  $q_{20}$  for the second link. Here are the numerical values of the input data.

**$m_1 = 1;$**   
 **$L_1 = 0.3;$**   
 **$m_2 = 1;$**   
 **$L_2 = 0.5;$**   
 **$d_c = 0.0254;$**   
 **$g = 9.8;$**   
 **$\rho = 2.46 \times 10^3;$**   
 **$q_{10} = 75 / 180 \text{ Pi};$**   
 **$q_{20} = 45 / 180 \text{ Pi};$**

This is the position vector from the origin  $O$  to the mass center  $C_1$  of the first link.

$$\mathbf{r}_{C_1} = L_1 / 2 \{ \text{Sin}[q_1[t]], 0, \text{Cos}[q_1[t]] \};$$

This is the position vector from the origin  $O$  to the point  $A$ .

$$\mathbf{r}_A = 2 \mathbf{r}_{C_1};$$

This is the position vector from the origin  $O$  to the mass center  $C_2$  of the second link.

$$\mathbf{r}_{C_2} = \mathbf{r}_A + L_2 / 2 \{ \text{Sin}[\mathbf{q}_2[\mathbf{t}]], 0, \text{Cos}[\mathbf{q}_2[\mathbf{t}]] \};$$

Here is the acceleration vector of the position vector.

$$\mathbf{a}_{C_2} = \mathbf{D}[\mathbf{r}_{C_2}, \{ \mathbf{t}, 2 \}];$$

We calculate the immersed depth  $z_T$  and the length  $L_E$ .

$$\begin{aligned} z_T &= L_1 \text{Cos}[\mathbf{q}_1[\mathbf{t}]] + L_2 \text{Cos}[\mathbf{q}_2[\mathbf{t}]] - (L_1 \text{Cos}[\mathbf{q}_{10}] + L_2 \text{Cos}[\mathbf{q}_{20}]); \\ L_E &= L_2 - 0.5 z_T / \text{Cos}[\mathbf{q}_2[\mathbf{t}]]; \end{aligned}$$

Here are the position vector and the velocity of the force application point E.

$$\begin{aligned} \mathbf{r}_E &= \mathbf{r}_A + L_E \{ \text{Sin}[\mathbf{q}_2[\mathbf{t}]], 0, \text{Cos}[\mathbf{q}_2[\mathbf{t}]] \}; \\ \mathbf{V}_E &= \mathbf{D}[\mathbf{r}_A, \mathbf{t}] + \text{Cross}[\mathbf{D}[\{0, \mathbf{q}_2[\mathbf{t}], 0\}, \mathbf{t}], (\mathbf{r}_E - \mathbf{r}_A)]; \end{aligned}$$

These are the gravitational forces acting at  $C_1$  and  $C_2$ .

$$\begin{aligned} \mathbf{G}_1 &= \{0, 0, m_1 \mathbf{g}\}; \\ \mathbf{G}_2 &= \{0, 0, m_2 \mathbf{g}\}; \end{aligned}$$

The function `Sign` can be used with `NDSolve` with the `FixedStep` method as mentioned previously. Therefore, we define the horizontal dynamic frictional force.

$$\begin{aligned} \beta &= 1569.7; \\ \mathbf{F}_{dh} &= -\text{Sign}[\mathbf{V}_E[[1]]] \beta d_c z_T \mathbf{V}_E[[1]]^2; \end{aligned}$$

Here is the vertical dynamic frictional force.

$$\mathbf{F}_{dv} = -\text{Sign}[\mathbf{V}_E[[3]]] \beta d_c z_T \text{Abs}[\text{Tan}[\mathbf{q}_2[\mathbf{t}]]] \mathbf{V}_E[[3]]^2;$$

Here is the horizontal static resistance force.

$$\begin{aligned} \eta_h &= 2.43; \\ \mathbf{F}_{sh} &= -\text{Sign}[\mathbf{V}_E[[1]]] \eta_h \rho g d_c z_T^2; \end{aligned}$$

We express the vertical resistance force coefficients,  $\eta_v = 10$  and  $\lambda = 1.4$  for plunging motion,  $\eta_v = 0.5$  and  $\lambda = 1.7$  for withdrawing motion.

$$\begin{aligned} \eta_v &= \text{Piecewise}[\{\{10, \mathbf{V}_E[[3]] > 0\}, \{0.5, \mathbf{V}_E[[3]] < 0\}\}]; \\ \lambda &= \text{Piecewise}[\{\{1.4, \mathbf{V}_E[[3]] > 0\}, \{1.7, \mathbf{V}_E[[3]] < 0\}\}]; \end{aligned}$$

This is the vertical static resistance force.

$$\mathbf{F}_{sv} = -\text{Sign}[\mathbf{V}_E[[3]]] \eta_v (z_T / d_c)^\lambda g \rho z_T / \text{Cos}[\mathbf{q}_2[\mathbf{t}]] d_c^2 \text{Pi} / 4;$$

This is the resistance force.

$$\mathbf{F}_r = \{\mathbf{F}_{dh} + \mathbf{F}_{sh}, \mathbf{0}, \mathbf{F}_{dv} + \mathbf{F}_{sv}\};$$

The ODE for the first link, where  $I_O$  is the mass moment inertia about the origin O, is

$$I_O \ddot{q}_1 = (\mathbf{r}_{C_1} \mathbf{G}_1 + \mathbf{r}_A \times (-m_2 \mathbf{a}_{C_2} + \mathbf{G}_2 + \mathbf{F}_r)) \cdot (0, 1, 0). \quad (6)$$

$$\mathbf{I}_O = m_1 / 3 L_1^2;$$

Eq6 =

$$\mathbf{I}_O \mathbf{q}_1''[t] == (\text{Cross}[\mathbf{r}_{C_1}, \mathbf{G}_1] + \text{Cross}[\mathbf{r}_A, (-m_2 \mathbf{a}_{C_2} + \mathbf{G}_2 + \mathbf{F}_r)])[[2]];$$

The ODE of the second link, where  $I_{C_2}$  is the mass moment inertia about  $C_2$ , is

$$I_{C_2} \ddot{q}_2 = ((\mathbf{r}_A - \mathbf{r}_{C_2}) \times (m_2 \mathbf{a}_{C_2} - \mathbf{G}_2 - \mathbf{F}_r) + (\mathbf{r}_E - \mathbf{r}_{C_2}) \times \mathbf{F}_r) \cdot (0, 1, 0). \quad (7)$$

$$\mathbf{I}_{C_2} = m_2 / 12 L_2^2;$$

Eq7 =

$$\mathbf{I}_{C_2} \mathbf{q}_2''[t] == (\text{Cross}[(\mathbf{r}_A - \mathbf{r}_{C_2}), (m_2 \mathbf{a}_{C_2} - \mathbf{G}_2 - \mathbf{F}_r)] + \text{Cross}[(\mathbf{r}_E - \mathbf{r}_{C_2}), \mathbf{F}_r])[[2]];$$

The ODEs are solved in `double`. The input variables `dq1` and `dq2` are the initial impact angular velocities of the pendulum. The initial impact angles, `q10`, `q20`, and the initial impact angular velocities, `dq10`, `dq20`, are defined as initial values. The solution is obtained using the `ExplicitRungeKutta` method with `FixedStep` with step size  $10^{-6}$  s and stops automatically using the `EventLocator` method when  $\dot{q}_1$  and  $\dot{q}_2$  simultaneously become zero (up to numerical error).

```
double[dq1_, dq2_] :=
Module[{q10 = q10, q20 = q20, dq10 = dq1, dq20 = dq2},
  stepsize = 10-6;
  eventFn = If[Abs[q1'[t]] < 2 × 10-3 && Abs[q2'[t]] < 2 × 10-3,
    0, 1];

  sol = NDSolve[{Eq6, Eq7, q1[0] == q10, q1'[0] == dq10,
    q2[0] == q20, q2'[0] == dq20}, {q1, q2}, {t, 0, 1},
    MaxSteps → Infinity, StartingStepSize → stepsize,
    MaxStepFraction → Infinity,
    Method → {"EventLocator", "Direction" → -1,
      "Event" → eventFn,
      "EventAction" → Throw[tend = t, "StopIntegration"],
      "EventLocationMethod" → "LinearInterpolation",
      "Method" → {"FixedStep",
        "Method" → "ExplicitRungeKutta"}}]; ]
```

The simulation of the model is performed for initial impact angles ( $q_{10} = 75^\circ$  and  $q_{20} = 45^\circ$ ) and for three initial impact angular velocities ( $\dot{q}_{10} = \dot{q}_{20} = -5, -3, -1$  rad/s).

The angle  $q$  and the angular velocity  $\dot{q}$  are plotted in blue for  $\dot{q}_{10} = \dot{q}_{20} = -5$  rad/s, red dashed for  $-3$  rad/s, and black dot-dashed for  $-1$  rad/s.

```

double[-5, -5];
Print["The stopping time for  $\dot{q}_{10} = \dot{q}_{20} = -5$  rad/s is ",
  tend, " [s]"];
SetOptions[Plot, PlotRange → Full, PlotStyle → Blue];
q1plot1 = Plot[Evaluate[q1[t] 180 / Pi /. sol], {t, 0, tend}];
q2plot1 = Plot[Evaluate[q2[t] 180 / Pi /. sol], {t, 0, tend}];
dq1plot1 = Plot[Evaluate[q1'[t] /. sol], {t, 0, tend}];
dq2plot1 = Plot[Evaluate[q2'[t] /. sol], {t, 0, tend}];

double[-3, -3];
Print["The stopping time for  $\dot{q}_{10} = \dot{q}_{20} = -3$  rad/s is ",
  tend, " [s]"];
SetOptions[Plot, PlotRange → Full, PlotStyle → {Red, Dashed}];
q1plot2 = Plot[Evaluate[q1[t] 180 / Pi /. sol], {t, 0, tend}];
q2plot2 = Plot[Evaluate[q2[t] 180 / Pi /. sol], {t, 0, tend}];
dq1plot2 = Plot[Evaluate[q1'[t] /. sol], {t, 0, tend}];
dq2plot2 = Plot[Evaluate[q2'[t] /. sol], {t, 0, tend}];

double[-1, -1];
Print["The stopping time for  $\dot{q}_{10} = \dot{q}_{20} = -1$  rad/s is ",
  tend, " [s]"];
SetOptions[Plot, PlotRange → Full,
  PlotStyle → {Black, DotDashed}];
q1plot3 = Plot[Evaluate[q1[t] 180 / Pi /. sol], {t, 0, tend}];
q2plot3 = Plot[Evaluate[q2[t] 180 / Pi /. sol], {t, 0, tend}];
dq1plot3 = Plot[Evaluate[q1'[t] /. sol], {t, 0, tend}];
dq2plot3 = Plot[Evaluate[q2'[t] /. sol], {t, 0, tend}];



Show[q1plot1, q1plot2, q1plot3, Frame → True,
  FrameLabel → {"time [s]", "q1 [°]"},
  PlotRange → {{0, tend}, {-44, 80}}, GridLines → Automatic]
Show[q2plot1, q2plot2, q2plot3, Frame → True,
  FrameLabel → {"time [s]", "q2 [°]"},
  PlotRange → {{0, tend}, {40, 72}}, GridLines → Automatic]
Show[dq1plot1, dq1plot2, dq1plot3, Frame → True,
  FrameLabel → {"time [s]", "q̇1 [rad/s]"},
  PlotRange → {{0, tend}, {-9, 0.4}}, GridLines → Automatic]
Show[dq2plot1, dq2plot2, dq2plot3, Frame → True,
  FrameLabel → {"time [s]", "q̇2 [rad/s]"},
  PlotRange → {{0, tend}, {-5.5, 7}}, GridLines → Automatic]


```

The stopping time for  $\dot{q}_{10} = \dot{q}_{20} = -5$  rad/s is 0.389631 [s]

The stopping time for  $\dot{q}_{10} = \dot{q}_{20} = -3$  rad/s is 0.457272 [s]

The stopping time for  $\dot{q}_{10} = \dot{q}_{20} = -1$  rad/s is 0.53981 [s]





The simulation results of the model for the initial impact angles ( $q_{10} = 75^\circ$  and  $q_{20} = 45^\circ$ ) show that the stopping time decreases when the initial impact velocity of the pendulum increases: for  $\dot{q}_{10} = \dot{q}_{20} = -5$  rad/s the stopping time is 0.389631 s, for  $\dot{q}_{10} = \dot{q}_{20} = -3$  rad/s the stopping time is 0.457272 s, and for  $\dot{q}_{10} = \dot{q}_{20} = -1$  rad/s the stopping time is 0.53981 s.

## ■ Conclusion

This article considers the impact of a kinematic chain into granular matter. The solution techniques are based on the resistance forces as a sum of a static (depth-dependent) resistance force and a dynamic (velocity-dependent) frictional force. We apply the `NDSolve` command to solve the equations of motion. For the single pendulum, `NDSolve` is used with `EventLocator` to identify when the velocity changes sign. The simulation is then stopped and a new simulation is performed using as initial conditions the end values of the previous simulation. For a double pendulum, `NDSolve` is used with the `FixedStep` method and the simulation is performed over the whole interval of time.

We observe that the numerical solution leads to a seemingly paradoxical result: as the initial velocity increases, the stopping time decreases.

## ■ Acknowledgments

For this work we used *Mathematica 7.0.1* and a Intel Core Duo computer running Mac OS X with a 2.0 GHz CPU and 2 GB of RAM.

## ■ References

- [1] H. Katsuragi and D. J. Durian, “Unified Force Law for Granular Impact Cratering,” *Nature Physics*, **3**, 2007 pp. 420–423. [www.nature.com/nphys/journal/v3/n6/full/nphys583.html](http://www.nature.com/nphys/journal/v3/n6/full/nphys583.html).
  - [2] L. S. Tsimring and D. Volfson, “Modeling of Impact Cratering in Granular Media,” in *Powders and Grains 2005, Proceedings of International Conference on Powders & Grains 2005* (R. García-Rojo, H. J. Herrmann, and S. McNamara, eds.), London: Taylor & Francis, 2005 pp. 1215–1218.
  - [3] R. Albert, M. A. Pfeifer, A.-L. Barabási, and P. Schiffer, “Slow Drag in a Granular Medium,” *Physical Review Letters*, **82**(1), 1999 pp. 205–208. [www.prl.aps.org/abstract/PRL/v82/i1/p205\\_1](http://www.prl.aps.org/abstract/PRL/v82/i1/p205_1).
  - [4] I. Albert, P. Tegzes, B. Kahng, R. Albert, J. G. Sample, M. Pfeifer, A.-L. Barabási, T. Vicsek, and P. Schiffer, “Jamming and Fluctuations in Granular Drag,” *Physical Review Letters*, **84**, 2000 pp. 5122–5125. [arxiv.org/abs/cond-mat/9912336](http://arxiv.org/abs/cond-mat/9912336).
  - [5] I. Albert, P. Tegzes, R. Albert, J. G. Sample, A.-L. Barabási, T. Vicsek, B. Kahng, and P. Schiffer, “Stick-Slip Fluctuations in Granular Drag,” *Physical Review E*, **64**, 2001 031307(9). [arxiv.org/abs/cond-mat/0102523](http://arxiv.org/abs/cond-mat/0102523).
  - [6] I. Albert, J. G. Sample, A. J. Morss, S. Rajagopalan, A.-L. Barabási, and P. Schiffer, “Granular Drag on a Discrete Object: Shape Effects on Jamming,” *Physical Review E*, **64**, 2001 061303(4). [arxiv.org/abs/cond-mat/0107392](http://arxiv.org/abs/cond-mat/0107392).
  - [7] M. B. Stone, R. Barry, D. P. Bernstein, M. D. Pelc, Y. K. Tsui, and P. Schiffer, “Studies of Local Jamming via Penetration of a Granular Medium,” *Physical Review E*, **70**, 2004 041301(10). [arxiv.org/abs/cond-mat/0404587](http://arxiv.org/abs/cond-mat/0404587).
  - [8] G. Hill, S. Yeung, and S. A. Koehler, “Scaling Vertical Drag Forces in Granular Media,” *Europhysics Letters*, **72**(1), 2005 pp. 137–142. [www.iopscience.iop.org/0295-5075/72/1/137](http://www.iopscience.iop.org/0295-5075/72/1/137).
  - [9] A. L. Yarin, M. B. Rubin, and I. V. Roisman, “Penetration of a Rigid Projectile into an Elastic-Plastic Target of Finite Thickness,” *International Journal of Impact Engineering*, **16**(5), 1995 pp. 801–831. [www.sciencedirect.com/science/article/pii/0734743X95000197](http://www.sciencedirect.com/science/article/pii/0734743X95000197).
  - [10] G. Yossifon, A. L. Yarin, and M. B. Rubin, “Penetration of a Rigid Projectile into a Multi-Layered Target: Theory and Numerical Computations,” *International Journal of Engineering Science*, **40**(12), 2002 pp. 1381–1401. doi:10.1016/S0020-7225(02)00013-7.
  - [11] S. N. Coppersmith, C.-H. Liu, S. Majumdar, O. Narayan, and T. A. Witten, “A Model for Force Fluctuations in Bead Packs,” *Physical Review E*, **53**(5), 1996 pp. 4673–4685. [www.arxiv.org/abs/cond-mat/9511105](http://www.arxiv.org/abs/cond-mat/9511105).
- S. Lee and D. B. Marghitu, “Impact of a Planar Kinematic Chain with Granular Matter,” *The Mathematica Journal*, 2011. [dx.doi.org/doi:10.3888/tmj.13-2](http://dx.doi.org/doi:10.3888/tmj.13-2).

### About the Authors

Seunghun Lee was a Ph.D. student in the mechanical engineering department at Auburn University, studying theoretical modeling of robotic systems. He is now serving in the Korean army.

Dan B. Marghitu is a professor of mechanical engineering at Auburn University. His research areas are impact dynamics, mechanisms, robots, and nonlinear dynamics.

*Mechanical Engineering Dept.  
Auburn University  
270 Ross Hall, Auburn, AL 36849  
leeseun@auburn.edu  
marghitu@auburn.edu*